tested 190817 using *SpinDynamica* 3.0.1 under *Mathematica* 11.0

spectrum of D2 in a liquid crystal

init

It is not obvious what the spectrum of D2 in a liquid crystal should look like. The two deuteriums have quadrupole couplings, a J-coupling, and a dipole-dipole coupling.

This problem was discussed in E. E. Burnell, C. A. de Lange and J. G. Snijders, "Nuclear magnetic resonance study of H2, HD, and D2 in nematic solvents," Physical Review A 25, 2339 (1982). The parameters are from table III of that paper.

TABLE III. Experimental coupling constants and order parameters. For HD, results from proton NMR are labeled H̲D and from deuterium NMR, HD̲.

Liquid crystal	Temperature K	Molecule	D^a (Hz)	B (Hz)	B/D^b	$S\times10^3$ (from D)
1167	310	H_2	1141.2±0.3			−3.967±0.001
		H̲D	186.02±0.15			−4.204±0.003
		HD̲	185.68±0.06	−655.86±0.06	−23.010 ±0.008	−4.196±0.001
		D_2	30.63±0.02	−706.98±0.05	−23.08 ±0.02	−4.497±0.003

```
Needs["SpinDynamica`"]
```

SpinDynamica version 3.0.1 loaded

ModifyBuiltIn: The following built-in routines have been modified in SpinDynamica:

{Chop, Dot, Duration, Exp, Expand, ExpandAll, NumericQ, Plus, Power, Simplify, Times, WignerD}.

Evaluate ??*symbol* to generate the additional definitions for *symbol*.

```
SetSpinSystem[{{1, 1}, {2, 1}}]
```

SetSpinSystem: the spin system has been set to {{1, 1}, {2, 1}}

SetBasis: the state basis has been set to ZeemanBasis[{{1, 1}, {2, 1}}, BasisLabels → Automatic].

```
BasisKets[]
```

{|1 +1, 1 +1⟩, |1 0, 1 +1⟩, |1 -1, 1 +1⟩, |1 +1, 1 0⟩,
|1 0, 1 0⟩, |1 -1, 1 0⟩, |1 +1, 1 -1⟩, |1 0, 1 -1⟩, |1 -1, 1 -1⟩}

```
H0 = ωJ opI[1].opI[2] + ωD (3 opI[1, "z"].opI[2, "z"] - opI[1].opI[2]) +
  (ωB / 3) Sum[3 opI[i, "z"].opI[i, "z"] - opI[i].opI[i], {i, 1, 2}]
```

$$\omega J\,(I_{1x}\cdot I_{2x} + I_{1y}\cdot I_{2y} + I_{1z}\cdot I_{2z}) + \omega D\,(-(I_{1x}\cdot I_{2x}) - I_{1y}\cdot I_{2y} + 2\,(I_{1z}\cdot I_{2z})) + \frac{1}{3}\,\omega B\,(-(I_{1x}\cdot I_{1x}) - I_{1y}\cdot I_{1y} + 2\,(I_{1z}\cdot I_{1z}) - I_{2x}\cdot I_{2x} - I_{2y}\cdot I_{2y} + 2\,(I_{2z}\cdot I_{2z}))$$

```
MatrixRepresentation[H0] // MatrixForm
```

$$\begin{pmatrix}
\frac{2\,\omega B}{3}+2\,\omega D+\omega J & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -\frac{\omega B}{3} & 0 & -\omega D+\omega J & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & \frac{2\,\omega B}{3}-2\,\omega D-\omega J & 0 & -\omega D+\omega J & 0 & 0 & 0 & 0 \\
0 & -\omega D+\omega J & 0 & -\frac{\omega B}{3} & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & -\omega D+\omega J & 0 & -\frac{4\,\omega B}{3} & 0 & -\omega D+\omega J & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & -\frac{\omega B}{3} & 0 & -\omega D+\omega J & 0 \\
0 & 0 & 0 & 0 & -\omega D+\omega J & 0 & \frac{2\,\omega B}{3}-2\,\omega D-\omega J & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & -\omega D+\omega J & 0 & -\frac{\omega B}{3} & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{2\,\omega B}{3}+2\,\omega D
\end{pmatrix}$$

```
MatrixRepresentation[H0] // MatrixPlot
```



calculations with J=0

```
parameters = {ωD → 2 π 30, ωB → -710 × 2 π, ωJ → 0}
```

{ωD → 60 π, ωB → -1420 π, ωJ → 0}

```
Tmax = 0.5;
```

- **simulate spectrum using Signal1D**

```
signal = Signal1D[{0, Tmax, Tmax / 1024},
  BackgroundGenerator → (H0 /. parameters)
 ]
```

Signal1D: Using SignalCalculationMethod → Diagonalization

Signal1D: the last sampling point has been dropped in order to get an even number of points.

Signal1D: Using LineBroadening → $2\pi \times 2.93174$ rad s^{-1}.

Signal[{0, $500. \times 10^{-3}$, 488.281×10^{-6}} , { Lorentzian ,<< 17 >>}]

```
ListPlot[
 Re[signal],
 Joined → True, PlotRange → All, Frame → True
]
```




```
ListPlot[
 Im[signal],
 Joined → True, PlotRange → All, Frame → True
]
```



```
ListPlot[
 Re@FT@Signal1D[{0, Tmax, Tmax / 1024},
    BackgroundGenerator → (H0 /. parameters)
   ],
 Joined → True, PlotRange → All, Frame → True
]
```

Signal1D: Using SignalCalculationMethod → Diagonalization

Signal1D: the last sampling point has been dropped in order to get an even number of points.

Signal1D: Using LineBroadening → 2π × 2.93174 rad s^{-1}.



calculations with J=100 Hz

```
parameters = {ωD → 2 π 30, ωB → -710 × 2 π, ωJ → 2 π 100}
```

```
{ωD → 60 π, ωB → -1420 π, ωJ → 200 π}
```

- **simulate spectrum using Signal1D**

```
ListPlot[
 Re@FT@Signal1D[{0, Tmax, Tmax / 1024},
    BackgroundGenerator → (H0 /. parameters)
   ],
 Joined → True, PlotRange → All, Frame → True
]
```

Signal1D: Using SignalCalculationMethod → Diagonalization

Signal1D: the last sampling point has been dropped in order to get an even number of points.

Signal1D: Using LineBroadening → $2\pi \times 2.93174$ rad s^{-1}.